



3. [bookmark: _Toc265195503]PROGRAMACION DEL ATMEGA8
; ******************************************************
; BASIC .ASM template file for AVR
[bookmark: _GoBack]; ******************************************************

.include "C:\VMLAB\include\m8def.inc"

.cseg
.org $00

RJMP INICIO

INICIO:

	LDI	R16, HIGH(RAMEND) 	; configura el puntero de pila
	OUT 	SPH, R16					; al final de la RAM
	LDI 	R16, LOW(RAMEND)
	OUT 	SPL, R16

	LDI 	R16, $FF					; configura el PD como salida   selector mux
	OUT 	DDRD, R16			
		
	RCALL CONFIGURA_USART		; configura el puerto serial

LDI ZH, HIGH(MENSAJE*2)       ; transmision del mensaje de inicio
LDI ZL, LOW(MENSAJE*2)
RCALL ENVIA_CADENA

ESPERA_X:
RCALL RX_DATO
CPI R20, 'X'						; si el dato ingresado es  'X'
BREQ LEER_SENX 					; empieza la lectura, conversion y transmision de los datos
RJMP ESPERA_X

LEER_SENX:
	RCALL ANALOGODIGITAL			; configura el conversor análogo digital
LEER_SEN:
LDI R17,0        					; R17 usado como selector del mux
LDI R22,0
LDI R18, 10
RCALL TX_DATO
LDI R18, 13
RCALL TX_DATO
LEER_SENSORES:
	MOV R17,R22
	LSL R17
   LSL R17
   LSL R17
   LSL R17
	OUT PORTD,R17
	RCALL LEEADC 					 ; el valor digital se encuentra en R16
	RCALL RETARDO					 ; retardo de 50ms para leer cada sensor
	RCALL Cambia_formato        ; cambia el valor digital y lo transmite en formato hexadecimal
	INC R22
	CPI R22,16
	BRNE LEER_SENSORES;
RJMP LEER_SEN

; **********************
; **** SUBRUTINAS ****
; **********************

;***************CONFIGURA USART*************************************

CONFIGURA_USART:

PUSH 	R16				; 9600 baudios, 8 bits, sin paridad,
LDI 	R16, $00			; 1 bit de parada, asincrono
OUT 	UBRRH, R16			
LDI 	R16, $C
OUT 	UBRRL, R16
LDI 	R16, $02
OUT 	UCSRA, R16
LDI 	R16, $18
OUT 	UCSRB, R16
LDI 	R16, $86
OUT 	UCSRC, R16
POP R16
RET

;***************CONFIGURA PUERTO ANALOGO DIGITAL*********************

ANALOGODIGITAL:

PUSH 	R16
LDI 	R16, 0B10000011
OUT 	ADCSR, R16
LDI 	R16, 0B01100000
OUT 	ADMUX, R16
POP 	R16
RET
	
;***************RECIBE DATO******************************************

RX_DATO:
	SBIS UCSRA, RXC
	RJMP RX_DATO
	IN R20, UDR
	RET
	
;***************TRANSMITE DATO***************************************	

TX_DATO:				; esta rutina envia el byte del registro r16 al
SBIS 	UCSRA,UDRE		; registro udr del usart
RJMP TX_DATO 		; espera a que udr este libre
OUT 	UDR,R18
RET	

;***************REALIZA LA CONVERSION ANALOGO-DIGITAL****************

LEEADC:
CBI 	ADMUX,0			; selecciona al canal ADC0
SBI 	ADCSR,6
ESPERACONVERSION:
SBIC 	ADCSR,6
RJMP ESPERACONVERSION
IN 	R16,ADCL
IN 	R16,ADCH
MOV R21, R16
RET

;*************** SUBRUTINA DE RETARDO *******************************
;*************** Retardo de aproximadamente 52mseg ******************
;*************** (199(12 + 25(10)) + 6) us **************************

RETARDO:
PUSH R16
PUSH R17
LDI R16, $C8 ; 1 ciclo
LAZO01:
LDI R17, $19 ; 1 ciclo
LAZO02:
NOP
NOP
NOP
NOP
NOP
NOP ; 6 ciclos
DEC R17 ; 1 ciclo
CPI R17, $00 ; 1 ciclo
BRNE LAZO02 ; 2 ciclos
NOP
NOP
NOP
NOP
NOP
NOP
DEC R16 ; 1 ciclo
CPI R16, $00 ; 1 ciclo
BRNE LAZO01 ; 2 ciclos
POP R17
POP R16
RET

;***************** CAMBIA DE FORMATO BINARIO A HEXADECIMAL**************

Cambia_formato:
PUSH R16
PUSH R17

MOV 	R17, R16
ANDI 	R16, 0B11110000
LSR 	R16
LSR 	R16
LSR 	R16
LSR	R16
MOV 	R18, R16
RCALL CONVIERTE
RCALL TX_DATO

ANDI 	R17, 0B00001111
MOV 	R18, R17
RCALL CONVIERTE
RCALL TX_DATO
LDI R18, $20
RCALL TX_DATO
POP R17
POP R16
RET

;**************** CONVIERTE A ASCII*****************************************
CONVIERTE:

CPI R18, 10
BRSH LETRA
ORI R18, $30
RJMP FIN
LETRA:
LDI R19, $37
ADC R18, R19

FIN:
RET

;******************TRANSMITE UNA CADENA DE CARACTERES***********************
ENVIA_CADENA:
LPM R18,Z+
CPI R18, 0
BREQ FIN_CADENA
RCALL TX_DATO
RJMP ENVIA_CADENA

FIN_CADENA:
RET
;***************************************************************************

MENSAJE:
.db   "    Pulse X si esta listo para empezar la evaluacion:" ,13,10,13
.db   "S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11 S12 S13 S14 S15 S16 ",13,10,0

TABLAS REGISTRASDAS EN LA PRUEBA AL PROTOTIPO


